
Für den Import von Daten aus einer Excel-Arbeitsmappe in eine Access-Tabelle wird normalerweise die Anweisung "DoCmd.TransferSpreadsheet" eingesetzt. Nachteil: Es können immer nur Daten vom ersten Tabellenblatt der Arbeitsmappe übernommen werden. Zwar lässt sich über den Parameter "Range" von "TransferSpreadsheet" anstelle eines Bereiches wie "A1:G27" auch ein benannter Bereich wie "Bestellungen" angeben, aber das setzt voraus, dass dieser Bereich zuvor explizit eingerichtet wurde - in vielen automatisch generierten Arbeitsmappen ist das jedoch nicht der Fall. Um Daten gezielt beispielsweise von einem Tabellenblatt "Bestellungen" aus dem Bereich "A1:R325" zu importieren, können Sie jedoch ein kaum bekanntes Feature von SQL-Abfragen nutzen. Eine übliche SQL-Abfrage sieht zum Beispiel so aus:

SELECT * FROM Kunden;

"Kunden" ist der Name einer Tabelle, aus der die Datensätze selektiert werden sollen. Statt eines Tabellennamens kann jedoch auch eine Verbindungszeichenfolge angegeben werden, der Informationen zu einer externen Datenquelle bereitstellt:

[Excel 8.0; HDR=NO; IMEX=1; Database=X:\Pfad\Datei.xls].[Bestellungen$A1:R325]"

Mit dieser Anweisung wird als Datenquelle eine Excel-Datei "Z:\Pfad\Datei.xls" und darin der Bereich "A1:R325" auf dem Tabellenblatt "Bestellungen" spezifiziert. Weitere Details folgen später. Darauf basierend kann nun eine SQL-Abfrage wie die folgende ausgebaut und für die Initialisierung eines Recordsets herangezogen werden:

strFileName = "Z:\Datenaustausch\Bestellungen.xls"
strCon = "[Excel 8.0;HDR=NO;IMEX=1;Database=" & _
         strFileName & "].[Bestellungen$A1:R325]"
strSQL = "SELECT T.* " & _
         "AS SheetData " & _
         "FROM " & strCon & _
         "AS T;"

Das Ergebnis ist ein Recordset, das alle Zeilen des betreffenden Bereiches, hier "Bestellungen!A1:R325", bereitstellt. Basierend auf dem Recordset können nun die Daten über einen zweiten Recordset in eine Access-Tabelle übertragen werden. Im Folgenden finden Sie die Prozedur "ExcelImportSQL", die diese Aufgabe übernimmt. Kopieren Sie sie zunächst in ein VBA-Modul:

Sub ExcelImportSQL()
  Dim I&, J&, NumRows&, NumCols&
  Dim strFileName As String
  Dim strTabName As String
  Dim strCon As String, strRange As String
  Dim strSQL As String
  Dim strDDL As String, strFieldName As String
  'DAO
  Dim db As DAO.Database
  Dim rsIn As DAO.Recordset
  Dim rsOut As DAO.Recordset

  On Error Resume Next
  Set db = CurrentDb()

  strTabName = "ExcelImportTest"
  DoCmd.Close acTable, strTabName, acSaveNo
  db.Execute "DROP TABLE " & strTabName

  On Error GoTo ErrorHandler
  strFileName = _
    "Z:\Datenaustausch\Bestellungen.xls"
  strRange = "Bestellungen$A1:R325"
  strCon = "[Excel 8.0;HDR=NO;IMEX=1;Database=" & _
           strFileName & "].[" & strRange & "]" 
  strSQL = "SELECT T.* " & _
           "AS SheetData " & _
           "FROM " & strCon & _
           "AS T;"

  Set rsIn = db.OpenRecordset(strSQL, _
                              dbOpenDynaset)
  rsIn.MoveLast
  NumRows = rsIn.RecordCount
  NumCols = rsIn.Fields.Count
  rsIn.MoveFirst

  strDDL = "CREATE TABLE " & strTabName & " ("
  For I = 0 To NumCols - 1
    strFieldName = "[" & rsIn.Fields(I).Value & "]"
    strDDL = strDDL & _
             strFieldName & " TEXT (100), "
  Next I
  strDDL = Left$(strDDL, Len(strDDL) - 2)
  strDDL = strDDL & ");"
  db.Execute strDDL, dbSQLPassThrough
  Set rsOut = db.OpenRecordset(strTabName, _
                               dbOpenDynaset)
  rsIn.MoveNext 'Zeile mit Feldnamen überspringen

  For I = 1 To NumRows - 1
    rsOut.AddNew
    For J = 0 To rsIn.Fields.Count - 1
      rsOut.Fields(J).Value = rsIn.Fields(J).Value
    Next J
    rsOut.Update
    rsIn.MoveNext
  Next I

ExitSub:
  On Error Resume Next
  rsIn.Close
  rsOut.Close

  Set rsOut = Nothing
  Set rsOut = Nothing
  Set db = Nothing

  DoCmd.OpenTable strTabName

  Exit Sub

ErrorHandler:
  MsgBox "Fehler " & CStr(Err) & ": " & Err.Description
  Resume ExitSub

End Sub

Setzen Sie über Extras-Verweise eine Referenz auf "Microsoft DAO 3.6 Object Library" und passen dann die Zuweisungen der folgenden Variablen gemäß Ihrer Gegebenheiten an:

strTabName: Name der Zieltabelle
strFileName: Pfad/Dateiname der Excel-Arbeitsmappe
strRange: Bereich, der importiert werden soll

Bitte beachten Sie, dass die erste Zeile des Bereiches zwingend Feldnamen bereitstellen muss, da darauf aufbauend per DDL (Data Definition Language) die Zieltabelle aufgebaut wird!

Im zweiten Teil dieses erläutern wir die genaue Funktionsweise der Prozedur "ExcelImportSQL" und erklären, welche Nachbearbeitung der Daten nach dem Import eventuell noch vorzunehmen ist.

Im ersten Teil dieses Tipps der Woche hatten wir die Prozedur "ExcelImportSQL" vorgestellt, mit der sich Daten aus beliebigen Tabellen und Bereichen einer Excel-Arbeitsmappe per SQL-Anweisung selektieren und in eine Access-Tabelle übertragen lassen. Im zweiten Teil geht es nun um die Funktionsweise dieser Lösung sowie eventuell notwendige Nachbearbeitungen.

Zunächst zur Funktionsweise: Der Import erfolgt im Beispiel in eine Tabelle "ExcelImportTest", die im ersten Schritt mit "DoCmd.Close" geschlossen und mit "DROP TABLE" gelöscht wird, um später gemäß der importierten Felder passgenau neu angelegt zu werden. Danach wird eine Verbindung zu dem zu importierenden Bereich eine Excel-Arbeitsmappe aufgebaut, der anschließend per Recordset "rsIn" adressiert werden kann. Der Bereich muss in der ersten Zeile Feldnamen bereitstellen, die ausgelesen, in einer DDL-Anweisung zusammengesetzt und als Basis einer "CREATE TABLE"-Anweisung verwendet werden. Das Ergebnis ist eine neue, leere Tabelle gemäß Variable "strTabName", die im Weiteren per Recordset "rsOut" adressiert wird. In einer Schleife werden dann alle Datensätze des Quell-Recordsets "rsIn" und pro Datensatz alle Felder durchlaufen und in den Ziel-Recordset "rsOut" übertragen. Danach werden alle Recordsets geschlossen und Objektvariablen dereferenziert. Den Abschluss bildet das Öffnen der generierten Tabelle.

Durch die Verwendung des Datentyps "Text" für alle Felder kann die Routine sehr flexibel und für den Import beliebiger Daten verwendet werden, diese Flexibilität erkauft man sich jedoch mit einer eventuell notwendigen Nachbearbeitung einzelner Felder, da Feldinhalte wie beispielsweise Datumsangaben nicht dem erwarteten Ergebnis entsprechen - Excel liefert diese als Zahl wie zum Beispiel "34579" und nicht im Format "TT.MM.JJJJ". Um hier wieder Datumsangaben anzuzeigen, erstellen Sie eine Aktualisierungsabfrage basierend auf der Import-Tabelle und aktualisieren die Datumsfelder mit folgendem Ausdruck:

=CStr(CDate([Feldname]))

Auf diese Weise wird die Zahl zunächst in ein Datum und dieses Datum dann in eine Zeichenkette umgesetzt, die im Feld gespeichert werden kann. Benötigen Sie die Daten als "echtes" Datum im Format "Datum/Zeit" stellen Sie diesen Feldtyp einfach im Tabellenentwurf ein - Access konvertiert die Datumsangaben dann automatisch beim Speichern der Änderungen. Gleiches gilt beispielsweise für Preise: Diese Werte liefert Excel als normale Fließkommazahlen zum Beispiel als "101,45". Stellen Sie hier den Feldtyp "Währung" im Tabellennetwurf ein und Access konvertiert die Daten beim Speichern der Änderungen automatisch ins Währungsformat, so dass beispielsweise "101,45 EUR" angezeigt wird.
